%% US Public Debt and Safe Asset Market Power
%% Jason Choi, Rishabh Kirpalani, and Diego Perez
%% Nov 24, 2024

%% Solve N=1 Cournot Equilibrium 

%----------------------------------------------------------------
% 0. Housekeeping
%----------------------------------------------------------------

close all

%----------------------------------------------------------------
% 1. Defining variables
%----------------------------------------------------------------

// Endogenous Variables
var bus brw rb rkstar rk krw_star kus_star krw kus kstar k wstar w c_rw c_us drdb vrw vus y dMrwdb;

// Exogenous Variables
var nnu oomega A Astar;

// Shocks
varexo eps_nnu eps_oomega eps_A;

// Paramters
parameters ggamma bbeta eeta llambda aalpha Astarbar Abar iiota iiota_star ddelta_rw ddelta_us N_1
nnu_bar oomega_bar rrho_nnu rrho_oomega ssigma_nnu ssigma_oomega rrho_A ssigma_A rrho_Astar ssigma_Astar
brw_me1 rb_me1 rkstar_me1 rk_me1 krw_star_me1 kus_star_me1 krw_me1 kus_me1 kstar_me1 k_me1
capKstar_me1 capK_me1 wstar_me1 w_me1 crw_me1 cus_me1 vrw_me1 vus_me1 dMrwdb_me1 y_me1;

%----------------------------------------------------------------
% 2. Calibration
%----------------------------------------------------------------

% // Parameters
ggamma = 2;
bbeta = 0.9886;
eeta = 0.545;
llambda = 1;
aalpha = 0.3;
Astarbar = 0.9254;
Abar = 0.8154;
iiota = 0.9070;
iiota_star = 0.7939;
ddelta_rw = 0.1;
ddelta_us = 0.1;
N_1 = 1;
nnu_bar = 0.0042;
oomega_bar = 0.0063;
rrho_nnu = 0.99;
ssigma_nnu = 0.01;
rrho_oomega = 0.95;
ssigma_oomega = 0.3;
rrho_A = 0.95;
ssigma_A = 0.02;
rrho_Astar = rrho_A;
ssigma_Astar = ssigma_A;

% // Analytic Steady State (Monopoly Equilibrium, N=2)
brw_me1 = (oomega_bar/(N_1^llambda*nnu_bar*(1+1/N_1*(eeta-1))))^(1/(eeta-1-llambda));
bus_me1 = 1/N_1*(oomega_bar/(N_1^llambda*nnu_bar*(1+1/N_1*(eeta-1))))^(1/(eeta-1-llambda));
rb_me1 = 1/bbeta - nnu_bar*(brw_me1)^(eeta-1) - 1;
rkstar_me1 = 1/bbeta + ddelta_rw - 1;
rk_me1 = 1/bbeta + ddelta_us - 1;
krw_star_me1 = ((aalpha*(1-iiota_star)*Astarbar*((1/bbeta+ddelta_rw-1)/(aalpha*iiota_star*Astarbar))^((aalpha*(1-iiota_star)-1)/(aalpha*(1-iiota_star))))/(1/bbeta+ddelta_us-1))^((aalpha*(1-iiota_star))/(1-aalpha));
kus_star_me1 = ((1/bbeta+ddelta_rw-1)/(aalpha*iiota_star*Astarbar))^(1/(aalpha*(1-iiota_star)))*krw_star_me1^((1-iiota_star*aalpha)/(aalpha*(1-iiota_star)));
krw_me1 = ((aalpha*iiota*Abar*((1/bbeta+ddelta_us-1)/(aalpha*(1-iiota)*Abar))^((aalpha*iiota-1)/(aalpha*iiota)))/(1/bbeta+ddelta_rw-1))^((aalpha*iiota)/(1-aalpha));
kus_me1 = ((1/bbeta+ddelta_us-1)/(aalpha*(1-iiota)*Abar))^(1/(aalpha*iiota))*krw_me1^((1-(1-iiota)*aalpha)/(aalpha*iiota));
kstar_me1 = krw_star_me1 + krw_me1;
k_me1 = kus_star_me1 + kus_me1;
capKstar_me1 = krw_star_me1^iiota_star*kus_star_me1^(1-iiota_star);
capK_me1 = krw_me1^(1-iiota)*kus_me1^iiota;
wstar_me1 = Astarbar*(1-aalpha)*(capKstar_me1)^aalpha;
w_me1 = Abar*(1-aalpha)*(capK_me1)^aalpha;
crw_me1 = wstar_me1 + (rkstar_me1-ddelta_rw)*kstar_me1 + nnu_bar/eeta*(brw_me1)^eeta + rb_me1*brw_me1;
cus_me1 = w_me1 + (rk_me1-ddelta_us)*k_me1 - oomega_bar/(1+llambda)*(bus_me1)^(1+llambda) - rb_me1*(bus_me1);
drdb_me1 = -nnu_bar*(eeta-1)*brw_me1^(eeta-2);
dMrwdb_me1 = 0;
nnu_me1 = nnu_bar;
oomega_me1 = oomega_bar;
A_me1 = Abar;
Astar_me1 = Astarbar;
vrw_me1 = crw_me1^(1-ggamma)/(1-ggamma)/(1-bbeta);
vus_me1 = cus_me1^(1-ggamma)/(1-ggamma)/(1-bbeta);
y_me1 = A_me1*(kus_me1^iiota*krw_me1^(1-iiota))^aalpha;

%----------------------------------------------------------------
% 3. Model
%----------------------------------------------------------------

model;

brw = bus*N_1;

c_rw^(-ggamma) = bbeta*c_rw(+1)^(-ggamma)*(nnu*brw^(eeta-1)+1+rb);
c_rw^(-ggamma) = bbeta*c_rw(+1)^(-ggamma)*(1-ddelta_rw+rkstar);
c_rw + kstar + brw = wstar + (1-ddelta_rw+rkstar(-1))*kstar(-1) + nnu(-1)/eeta*(brw(-1))^eeta + (1+rb(-1))*brw(-1);

c_us^(-ggamma) = bbeta*c_us(+1)^(-ggamma)*(oomega*(bus)^(llambda)+1+rb+drdb*bus);
c_us^(-ggamma) = bbeta*c_us(+1)^(-ggamma)*(1-ddelta_us+rk);
c_us + k - bus = w + (1-ddelta_us+rk(-1))*k(-1) - oomega(-1)/(1+llambda)*(bus(-1))^(1+llambda) - (1+rb(-1))*bus(-1);

rk = Astar*aalpha*(1-iiota_star)*krw_star^(aalpha*iiota_star)*kus_star^(aalpha*(1-iiota_star)-1);
rkstar = Astar*aalpha*iiota_star*krw_star^(aalpha*iiota_star-1)*kus_star^(aalpha*(1-iiota_star));
rk = A*aalpha*iiota*krw^(aalpha*(1-iiota))*kus^(aalpha*iiota-1);
rkstar = A*aalpha*(1-iiota)*krw^(aalpha*(1-iiota)-1)*kus^(aalpha*iiota);
wstar = Astar*(1-aalpha)*krw_star^(aalpha*iiota_star)*kus_star^(aalpha*(1-iiota_star));
w = A*(1-aalpha)*krw^(aalpha*(1-iiota))*kus^(aalpha*iiota);

0 = -(drdb+nnu*(eeta-1)*brw^(eeta-2))*(c_rw(+1)/c_rw)^(-ggamma)+(1+rb+nnu*brw^(eeta-1))*dMrwdb;
0 = dMrwdb*(rkstar+1-ddelta_rw);

k = kus + kus_star;
kstar = krw + krw_star;

log(nnu) = (1-rrho_nnu)*log(nnu_bar) + rrho_nnu*log(nnu(-1)) + ssigma_nnu*eps_nnu;
log(oomega) = (1-rrho_oomega)*log(oomega_bar) + rrho_oomega*log(oomega(-1)) + ssigma_oomega*eps_oomega;
log(A) = (1-rrho_A)*log(Abar) + rrho_A*log(A(-1)) + ssigma_A*eps_A;
log(Astar) = (1-rrho_Astar)*log(Astarbar) + rrho_Astar*log(Astar(-1)) + ssigma_Astar*eps_A;

vrw = c_rw^(1-ggamma)/(1-ggamma) + bbeta*vrw(+1);
vus = c_us^(1-ggamma)/(1-ggamma) + bbeta*vus(+1);

y = A*(kus^iiota*krw^(1-iiota))^aalpha;

end;

%----------------------------------------------------------------
% 4. Computation
%----------------------------------------------------------------

initval;
  bus = bus_me1;
  brw = brw_me1;
  rb = rb_me1;
  rkstar = rkstar_me1;
  rk = rk_me1;
  krw_star = krw_star_me1;
  kus_star = kus_star_me1;
  krw = krw_me1;
  kus = kus_me1;
  kstar = kstar_me1;
  k = k_me1;
  wstar = wstar_me1;
  w = w_me1;
  c_rw = crw_me1;
  c_us = cus_me1;
  drdb = drdb_me1;
  dMrwdb = dMrwdb_me1;
  nnu = nnu_me1;
  oomega = oomega_me1;
  A = A_me1;
  Astar = Astar_me1;
  vrw = vrw_me1;
  vus = vus_me1;
  y = y_me1;
end;

resid;
check;

shocks;
  var eps_nnu = 1;
  var eps_oomega = 1;
  var eps_A = 1;
end;

set_dynare_seed('default');
stoch_simul(order=2,periods=10000,drop=0,nograph,noprint,pruning);

%----------------------------------------------------------------
% 5. Simulate transition
%----------------------------------------------------------------

spread_path = (rk-ddelta_us-rb)*100;
var_sp = var(spread_path);
auto_sp = autocorr(spread_path);
var_by = var(brw./y);
auto_by = autocorr(brw./y);
corr_pq_by = corr(spread_path,brw./y);

moments = [mean(brw./y) mean(spread_path) var_by var_sp corr_pq_by auto_by(2) auto_sp(2)]';
data_mom = [0.41 0.62 0.03 0.086 -0.56 0.96 0.70]';
rowNames = {'Mean b/y','Mean sp','Var b/y','Var sp','Corr (b/y,sp)','Autocorr b/y','Autocorr sp'};
colNames = {'Model Moments','Data Moments'};
TableA0 = array2table([moments data_mom],'RowNames',rowNames,'VariableNames',colNames)

%----------------------------------------------------------------
% 6. Calculate welfare from transition
%---------------------------------------------------------------

b_sme1 = mean(brw./y);
bus_sme1 = mean(bus./y);
spread_sme1 = mean(spread_path);
rb_sme1 = mean(rb)*100;

oo_me1 = oo_;
M_me1 = M_;
options_me1 = options_;  

save me1_save oo_me1 vus_me1 vrw_me1 cus_me1 crw_me1 M_me1 options_me1 b_sme1 bus_sme1 spread_sme1 rb_sme1;